Longitudinal Tests of Profile Similarity and Latent Transition Analyses¹

Alexandre J.S. Morin, Department of Psychology, Concordia University, Canada & **David Litalien**, Faculté des sciences de l'éducation, Université Laval, Canada

Cite as:

Morin, A.J.S., & Litalien, D. (2017). Webnote: Longitudinal Tests of Profile Similarity and Latent Transition Analyses. Montreal, QC: Substantive Methodological Synergy Research Laboratory.

Morin, Meyer, Creusier and Biétry (2016) proposed a methodological framework to systematically investigate the extent to which latent profile solutions generalize to different groups or samples of participants according to a sequence of profile similarity tests: (a) configural similarity (the same number of profiles based on the same indicators); (b) structural similarity (the same within-profile means); (c) dispersion similarity (the same within-profile variances); (d) distributional similarity (the same marginal probabilities, or relative size of the profiles). Following these initial steps, and starting from the most "similar" model, they further proposed to test the similarity of associations between profiles and covariates through tests of: (e) predictive similarity (relations between predictors and profile membership); (f) explanatory similarity (relations between profile membership and outcomes). Despite the fact that their contribution was limited to the multiple group context, they also proposed, as part of their online supplements, a similar methodology and relevant input files for longitudinal tests of profile similarity using Latent Transition Analyses (LTA). Similar methods and input files were also provided in chapters by Morin (2016) and Morin and Wang (2016), who extended this approach to multi-group tests of profile similarity using mixture regression models. More recently, Ciarrochi, Morin, Sahdra, Litalien, and Parker (2017) provided the first application of longitudinal tests of profile similarity using LTA, and even improved on the approach advocated before for tests of predictive similarity, which we summarize later in this webnote.

However, the approach described in these previous publications for tests of distributional similarity conducted in the LTA context remains suboptimal. More precisely, a LTA model is more than a simple longitudinal latent profile analysis (LPA) model. In a simple longitudinal LPA, distinct LPA models, relying on the same indicators measured at different time points, can be jointly estimated with participants freely allowed to have membership in any of the profiles at any of the time points. Such a model is akin to a longitudinal confirmatory factor analytic (CFA) model in which distinct factors are estimated at each time points using repeated measures of the same set of indicators. In this longitudinal CFA model, the factors estimated at Time 1 (F1) are simply allowed to correlate with the factors estimated at Time 2 (F2). In contrast, in a LTA model, the profiles estimated at Time 2 (c2) are predicted by, or regressed on, the profiles estimated at Time 1 (c1), leading to the inclusion, in Mplus, of the "c2 on c1;" statement in the "%OVERALL%" section of the model. This is thus more similar to an autoregressive longitudinal SEM model in which F1 are allowed to predict F2, rather than to a simple longitudinal CFA model where F1 and F2 are simply correlated.

The previously cited publications relied on LTA model specifications, and tested distributional similarity by implementing longitudinal equality constraints on the class probabilities while ignoring the regression component of the LTA model. To retain the CFA analogy, this is like constraining the mean of F1 to be equal to the intercept of the F2 on F1 regression. As such, they constrained the conditional probabilities of c2 (conditional on the c2 on c1 prediction) to be equal to the marginal probabilities of c1 (the effective size of the profiles at Time 1). A proper test of distributional similarity

¹ We want to acknowledge the invaluable help provided by the Mplus team (Linda and Bengt Muthén, as well as Tihomir Asparouhov) in the preparation of this webnote. Readers interested in learning more about these issues should consult the following documents available on the Mplus website and which have been very helpful in the preparation of the present document:

[•] Pages 557-561 in Chapter 14 of the Mplus 8 Manual

[•] Slides 47-50 in the Topic 6 Short Course Handout

[•] Mplus Webnote 13: LTA in Mplus: Transition probabilities influenced by covariates

FAQ: <u>LTA with Movers-Stayers</u> & <u>Class probabilities held equal across time</u>

[•] FAQ: LTA with transition probs varying as a function of covariates

would rather require that the marginal probabilities of c2 be constrained to be equal to the marginal probabilities of c1. Although it is possible to implement such constraints in LTA by way of an extensive set of model constraints (which we illustrate in the Appendix of this webnote, and is also illustrated in Class probabilities held equal across time using a probability parameterization), this approach adds considerably to the computational complexity of the model, and as such often results in model nonconvergence, increased difficulties of convergence on a true maximum likelihood, and substantially increases computational time.

However, it is not necessary to rely on LTA for tests of longitudinal profile similarity, given that such tests only require a longitudinal LPA model (e.g., similar to the longitudinal CFA model, whereas the LTA model is more similar to an autoregressive SEM). In the present webnote, we thus summarize the complete optimized sequence of longitudinal tests of profile similarity relying on a longitudinal LPA for steps a to d, and then converting the most similar model from this initial sequence to LTA for the remaining steps e and f. We also propose two alternative solutions to the commonly encountered problem of shifts in profile definitions following the incorporation of covariates to the model.

To review the full sequence, imagine a longitudinal LPA model including 3 profile indicators X1-X3, one predictor Y, and one outcome W, all assessed at two time points: (1) Time 1: X11-X31 Y1 W1; (2) Time 2: X12-X32 Y2 W2. The sequence of profile similarity tests starts with the model of configural similarity, where the same number of profiles, defined from the same indicators, are estimated at each time point. For illustration purposes, we provide the required Mplus syntax for a 3-profile solution.

DATA, VARIABLE, ANALYSES and OUTPUT sections of all models

```
DATA: FILE = data.dat;

VARIABLE:

NAMES = ID X11 X21 X31 Y1 W1 X12 X22 X32 Y2 W2;

IDVAR = ID; USEV = X11 X21 X31 X12 X22 X32;

CLASSES = c1 (3) c2 (3);

ANALYSIS:

TYPE = MIXTURE; ESTIMATOR = MLR;

STARTS = 10000 500; STITERATIONS = 1000;

OUTPUT:

STDYX SAMPSTAT CINTERVAL RESIDUAL SVALUES TECH1 TECH7 TECH15;
```

MODEL 1 (LPA, Step a): Configural Similarity (MODEL section)

```
MODEL:
%OVERALL%
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3);
X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6);
X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9);
X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (mm1-mm3);
X12 X22 X32 (vv1-vv3);
%C2#2%
[X12 X22 X32] (mm4-mm6);
X12 X22 X32 (vv4-vv6);
%C2#3%
[X12 X22 X32] (mm7-mm9);
X12 X22 X32 (vv7-vv9);
```

MODEL 2 (LPA, Step b): Structural Similarity (MODEL section)

```
MODEL:
%OVERALL%
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3);
X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6);
X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9);
X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3);
X12 X22 X32 (vv1-vv3);
%C2#2%
[X12 X22 X32] (m4-m6);
X12 X22 X32 (vv4-vv6);
%C2#3%
[X12 X22 X32] (m7-m9);
X12 X22 X32 (vv7-vv9);
```

MODEL 3 (LPA, Step c): Dispersion Similarity (MODEL section)

```
MODEL:
%OVERALL%
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3);
X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6);
X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9);
X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3);
X12 X22 X32 (v1-v3);
%C2#2%
[X12 X22 X32] (m4-m6);
X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9);
X12 X22 X32 (v7-v9);
```

MODEL 4 (LPA, Step d): Distributional Similarity (MODEL section)

```
MODEL:
%OVERALL%
[c1#1] (p1);
[c1#2] (p2);
[c2#1] (p1);
[c2#2] (p2);
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3);
X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6);
X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9);
X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3);
X12 X22 X32 (v1-v3);
%C2#2%
[X12 X22 X32] (m4-m6);
X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9);
X12 X22 X32 (v7-v9);
```

Once this stage has been reached, the final most similar model is retained for further stages of analyses and can easily be converted to a LTA model should one desire to investigate longitudinal profile transitions. However, the following steps differ based on whether distributional similarity has been supported or not, given the complexity of implementing distributional similarity constraints in LTA.

A. DISTRIBUTIONAL SIMILARITY IS REJECTED

Assuming that the last model of distributional similarity has been rejected, which means that the marginal probabilities or relative size of the profiles differ, the input files previously provided remain appropriate (LPA MODELS 1 to 3). Thus, assuming that the model of dispersion similarity was retained (MODEL 3), we can first convert it to a LTA model:

MODEL 5 (LTA; conversion between steps a-d and e-f): LTA of Dispersion Similarity

```
MODEL:
%OVERALL%
c2 on c1:
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3);
X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6);
X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9);
X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3);
X12 X22 X32 (v1-v3);
%C2#2%
[X12 X22 X32] (m4-m6);
X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9);
X12 X22 X32 (v7-v9);
```

Predictors can then be added to this model. Morin et al. (2016) proposed to contrast a model in which the effects of the predictors on profile membership were freely estimated across time points, to a model where the effects of these predictors were constrained to be equal across time points. Ciarrochi et al. (2017) noted that the interest of contrasting these two models lies in the fact that the model in which prediction are freely estimated across time points provide a direct estimate of the effects of predictors on longitudinal profile transitions (c1 to c2) given the fact that the effects of predictors on the c2 profiles is estimated over and above the effects of the c1 profiles on c2 profiles (over and above profile similarity). However, they also noted that it could be worth adding an additional (preliminary) step to this sequence to see whether the effects of predictors on c2 profile membership themselves differed as a function of membership into specific c1 profiles, i.e. whether these effects were moderated by profile membership. This revised sequence leads to three alternative models.

MODEL 6 (LTA, Step e.1): LTA with predictors' effects freely estimated across time points and Time 1 profiles

```
MODEL:
%OVERALL%
c2 on c1;
c1 ON Y1;
c2 ON Y2;
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3); X11 X21 X31 (v1-v3);
c2 ON Y2;
%C1#2%
[X11 X21 X31] (m4-m6); X11 X21 X31 (v4-v6);
c2 ON Y2;
%C1#3%
[X11 X21 X31] (m7-m9); X11 X21 X31 (v7-v9);
c2 ON Y2;
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3); X12 X22 X32 (v1-v3);
[X12 X22 X32] (m4-m6); X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9); X12 X22 X32 (v7-v9);
```

MODEL 7 (LTA, Step e.2): LTA with predictors' effects freely estimated across time points

```
MODEL:
%OVERALL%
c2 on c1;
c1 ON Y1;
c2 ON Y2;
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3); X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6); X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9); X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3); X12 X22 X32 (v1-v3);
%C2#2%
[X12 X22 X32] (m4-m6); X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9); X12 X22 X32 (v7-v9);
```

MODEL 8 (LTA, Step e.3): LTA with predictors: Predictive Similarity

```
MODEL:
%OVERALL%
c2 on c1;
c1 ON Y1 (pr1-pr2);
c2 ON Y2 (pr1-pr2);
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3); X11 X21 X31 (v1-v3);
%C1#2%
[X11 X21 X31] (m4-m6); X11 X21 X31 (v4-v6);
%C1#3%
[X11 X21 X31] (m7-m9); X11 X21 X31 (v7-v9);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3); X12 X22 X32 (v1-v3);
%C2#2%
[X12 X22 X32] (m4-m6); X12 X22 X32 (v4-v6);
%C2#3%
[X12 X22 X32] (m7-m9); X12 X22 X32 (v7-v9);
```

Outcomes can also be added to the retained model. Morin et al. (2016) proposed to contrast a model in which the relations between profile membership and the outcomes were freely estimated across time points, to a model where these relations were constrained to be equal across time points. This approach relies on the "MODEL CONSTRAINT" function to systematically test the magnitude and statistical significance of means differences across profiles and time points.

MODEL 9 (LTA, Step f.1): LTA with outcomes freely estimated across time points and profiles

```
MODEL:
%OVERALL%
c2 on c1;
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3); X11 X21 X31 (v1-v3);
[W1] (oa1);
%C1#2%
[X11 X21 X31] (m4-m6); X11 X21 X31 (v4-v6);
[W1] (oa2);
%C1#3%
[X11 X21 X31] (m7-m9); X11 X21 X31 (v7-v9);
[W1] (oa3);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3); X12 X22 X32 (v1-v3);
[W2] (ob1);
%C2#2%
[X12 X22 X32] (m4-m6); X12 X22 X32 (v4-v6);
[W2] (ob2);
%C2#3%
[X12 X22 X32] (m7-m9); X12 X22 X32 (v7-v9);
[W2] (ob3);
MODEL CONSTRAINT:
NEW (y12);
y12 = oa1-oa2;
NEW (y13);
y13 = oa1-oa3;
NEW (y23);
y23 = oa2-oa3;
NEW (z12);
z12 = ob1-ob2;
NEW (z13);
z13 = ob1-ob3:
NEW (z23);
z23 = ob2-ob3;
```

```
MODEL:
%OVERALL%
c2 on c1:
MODEL c1:
%C1#1%
[X11 X21 X31] (m1-m3); X11 X21 X31 (v1-v3);
[W1] (oa1);
%C1#2%
[X11 X21 X31] (m4-m6); X11 X21 X31 (v4-v6);
[W1] (oa2);
%C1#3%
[X11 X21 X31] (m7-m9); X11 X21 X31 (v7-v9);
[W1] (oa3);
MODEL c2:
%C2#1%
[X12 X22 X32] (m1-m3); X12 X22 X32 (v1-v3);
[W2] (oa1);
%C2#2%
[X12 X22 X32] (m4-m6); X12 X22 X32 (v4-v6);
[W2] (oa2);
%C2#3%
[X12 X22 X32] (m7-m9); X12 X22 X32 (v7-v9);
[W2] (oa3);
MODEL CONSTRAINT:
NEW (y12);
y12 = oa1-oa2;
NEW (y13);
y13 = oa1-oa3;
NEW (y23);
y23 = oa2-oa3;
```

B. PROFILE SHIFTING: SOLUTION 1

In theory, profile definition should be independent of predictions, correlates, and outcomes and, as such, the final model of longitudinal similarity retained as a result of the comparison of MODELS 1 to 4 should remain unchanged when moving to MODELS 5 to 10. Drastic changes are unlikely in MODEL 5 when distributional similarity is rejected, but they may still happen. An easy solution to this problem (Solution 1) is to use the starts values associated with the final retained model (MODELs 1 to 3 at least, as an alternative solution will be proposed later for situations of distributional similarity) in the estimation of MODELS 5 to 10. These starts values can be obtained by requesting **SVALUES** in the **OUTPUT** section of Mplus, and then turning the random starts function to 0 (**STARTS = 0**;) in the estimation of MODELs 5 to 10. This procedure should make it highly unlikely that any meaningful shifts in profile definition or size will occur when the model is converted to a LTA model with or without predictors or outcomes. Relying on grand-mean centred predictors is also likely to help avoid shifts in profile meaning. When those shifts still occur, then Solution 2 may prove to be more relevant.

C. DISTRIBUTIONAL SIMILARITY IS RETAINED & SOLUTION 2 TO PROFILE SHIFTS

When distributional similarity is retained, it becomes impractical (or at least more complex: see the Appendix to this webnote) to convert a longitudinal LPA model to a LTA in which the distributional similarity is maintained. In this context, MODELS 5 to 10 are best estimated using the manual implementation of the auxiliary 3-step approach described by Asparouhov and Muthén (2014). This approach also provides a solution to the profile shifting problem described above as it guarantees that, in most situations, the profile definition will remain unchanged.

Essentially, this approach consists of assigning participants to their most likely profile, while taking into account the probability that each participant has of being a member of each latent profile in the estimation of the model. To achieve this, the approach relies on the exact parameter estimates obtained from the most similar model from the preceding sequence (in this example, Distributional Similarity). These parameter estimates can be obtained, ready to cut and paste in a new input, while using the **SVALUES** function of the **OUTPUT** command.

Using these exact values (these would need to be fixed with @, rather than used as start values with *), one model is estimated separately for each time point and used to export the class membership information to an external data file. Asparouhov and Muthén (2014) further describe a process using the Auxiliary function through which the variables required for further analyses can be directly exported to this external data file, which can then be automatically used in further analyses. However, doing so poses problem in longitudinal analyses with missing data as the time-specific models will only be estimated using participants who completed the specific time point, so that the final data set will automatically use listwise deletion of participants who have not completed both measurement point. Our recommendation is thus to use your preferred data management package (e.g., SPSS, EXCEL) to combine the saved class membership information into the complete data set.

In these analyses, and all further analyses, the random start function is set to 0 (STARTS = 0;) to ensure that the model converges on the same solution as retained before.

Sample input for the Time 1 LPA model used to generate the class membership and probability information (the used starts values are fictive).

```
CLASSES = c1 (3);
Analysis:
TYPE = MIXTURE;
ESTIMATOR = MLR;
STARTS = 0;
MODEL:
%OVERALL%
[c1#1@0.34455](p1); [c1#2@-0.90364](p2);
[X11@-0.25119] (m1); [X12@-0.14630] (m2); [X13@-0.30598] (m3);
X11@1.17077 (v1); X12@0.81969 (v2); X13@0.69377 (v3);
%C1#2%
[X11@0.87814] (m4); [X12@0.13189] (m5); [X13@0.79368] (m6);
X11@0.03006 (v4); X12@0.60919 (v5); X13@0.30495 (v6);
%C1#3%
[X11@0.07476] (m7); [X12@0.17225] (m8); [X13@0.14160] (m9);
X11@0.06493 (v7); X12@0.35436 (v8); X13@0.31784 (v9);
SAVEDATA: file=c1.dat; save=cprob;
```

The end of the output file will indicate the name, and order, of the variables that are saved into C1.dat. SAVEDATA INFORMATION

```
Save file
c1.dat
Order and format of variables
X11 F10.3
X12 F10.3
[...]
CPROB1 F10.3
CPROB2 F10.3
CPROB3 F10.3
C1 F10.3
```

The variable C1 refers to the most likely class membership of each participant. This is the variable that needs to be brought back into the main data file, where it is relabelled N1 to avoid confusion with the way Mplus labels latent profiles. The same process is then repeated for Time 2 (and all other time points, whenever necessary). Using this information (N1 and N2), it then becomes possible to build an input file that will automatically, and systematically, result in the same classification of participants, while taking into account the class probabilities. More precisely, these variables are used directly in the model estimation process to force the assignment of participants into each profile.

Before doing so, however, we need further information from the outputs associated with the models that we just estimated to obtain the class probabilities and class membership associated separately with the Time 1 and Time 2 solutions. Just before the MODEL RESULTS section, at the end of the MODEL FIT INFORMATION section, there is a Table of logits for Time 1:

Logits for the Classification Probabilities for the Most Likely Latent Class Membership (Column) by Latent Class (Row)

```
1 2 3
1 4.002 2.045 0.000
2 1.403 2.765 0.000
3 -4.494 -2.119 0.000
```

And the same for Time 2:

```
1 2 3
1 5.496 3.718 0.000
2 2.014 3.454 0.000
3 -3.763 -2.406 0.000
```

These logits (at each measurement point) reflect the class probabilities of assignment of the participants into the various profiles and are used in the input to "correct" the class assignment of the participants into each profile while taking into account their probability of membership into the other profiles. This process is akin to modeling class assignment while controlling for the imprecise nature of this assignment, a process similar to controlling for measurement errors.

On next page is the input used to provide an exact replication of the model of Distributional Similarity and converting it to a LTA model (Corresponding to MODEL 5). This same set up can then be used to estimate MODELS 6 to 10, without having to worry about profile shifts.

It should be noted that the models estimated using this approach will not result in the same "model-based" estimates of marginal class probabilities as the model used to generate the 3-step approach. More precisely, tests of distributional similarity are conducted by constraining the "model-based" estimates of marginal class probability to equality across time points. However, the assignment of participants to profiles based either on their posterior probabilities of class membership or on their most likely class membership is never as accurate as the model-based estimates themselves. The 3-step approach relies on these class probabilities to correct the most likely class membership information. As such, it will result in similar estimates of marginal class probabilities based on the model-estimated posterior probabilities or most likely class assignment as the original model and can thus, for practical purposes, be considered to be equivalent.

```
DATA: FILE = c1c2.csv;
VARIABLE:
NAMES = ID n1p1 n1p2 n1p3 N1
n2p1 n2p2 n2p3 N2;
USEVARIABLES = N1 N2;
Nominal = N1 N2;
IDVAR = ID;
CLASSES = c1 (3) c2 (3);
ANALYSIS:
TYPE = MIXTURE;
ESTIMATOR = MLR;
Starts = 0;
MODEL:
%OVERALL%
Model C1:
%c1#1%
[n1#1@4.002]; [n1#2@2.045];
%c1#2%
[n1#1@1.403]; [n1#2@2.765];
%c1#3%
[n1#1@-4.494]; [n1#2@-2.119];
MODEL c2:
%c2#1%
[n2#1@5.496]; [n2#2@3.718];
%c2#2%
[n2#1@2.014]; [n2#2@3.454];
%c2#3%
[n2#1@-3.763]; [n2#2@-2.406];
```

References

- Asparouhov, T., & Muthén, B.O. (2014). Auxiliary variables in mixture modeling: Three-step approaches using Mplus. *Structural Equation Modeling*, 21, 1-13.
- Ciarrochi, J., Morin, A.J.S., Sahdra, B., Litalien, D., & Parker, P.D. (2017, In press). A Longitudinal person-centered perspective on youth social support: Relations with psychological wellbeing. *Developmental Psychology*.
- Morin, A.J.S. (2016). Person-centered research strategies in commitment research. In J.P. Meyer (Ed.), The Handbook of Employee Commitment (p. 490-508). Cheltenham, UK: Edward Elgar.
- Morin, A.J.S., & Wang, J.C.K. (2016). A gentle introduction to mixture modeling using physical fitness data. In N. Ntoumanis, & N. Myers (Eds.), *An Introduction to Intermediate and Advanced Statistical Analyses for Sport and Exercise Scientists* (pp. 183-210). London, UK: Wiley.
- Morin, A.J.S., Meyer, J.P., Creusier, J., & Biétry, F. (2016). Multiple-group analysis of similarity in latent profile solutions. *Organizational Research Methods*, 19 (2), 231-254.

APPENDIX. DISTRIBUTIONAL SIMILARITY IN A LTA MODEL

By default, Mplus relies on a logit parameterisation in the context of LTA. In the Mplus manual, toward the end of chapter 14, details about this parameterization are exposed. Taking the example of a LTA including 3-profiles at each time points, here is the expanded %Overall% section, in which the latent categorical variable representing the profiles at Time 1 (c1) is allowed to predict the latent categorical variable representing the profiles estimated at Time 2 (c2). Rather than using the short form "C2 ON C1", it is necessary, for test of distributional similarity using the full LTA parameterization, to use the complete presentation of the profile-to-profile regressions representing the conditional distribution of c2 given c1. As usual, the last latent profile is used as the referent, and does not need to be specified. The statements [c1#1] and [c1#2] refer to the relative sizes of profiles 1 and 2 at Time 1, whereas the statements [c2#1] and [c2#2] refer to the intercepts of the regressions of c2 on c1. A parameter label is also used (defined in parenthesis) for each of these statements, leading to the following %overall% section:

```
%OVERALL%
c2#1 ON c1#1 (b11);
c2#1 ON c1#2 (b12);
c2#2 ON c1#1 (b21);
c2#2 ON c1#2 (b22);
[c1#1] (p1); [c1#2] (p2);
[c2#1] (a1); [c2#2] (a2);
```

The Mplus manual lists, by their labels, the parameters combinations reflecting the log odds of membership to one specific c1 class relative to the last (referent c2 class), for a specific c1 class.

	C2				
		1	2	3	
C1	1	a1 + b11	a2 + b21	0	
	2	a1 + b12	a2 + b22	0	
	3	al	a2	0	

Using the information provided in this table, the MODEL CONSTRAINT function can be used to constrain the relative size (aka marginal probabilities) of the Time 2 profiles to match exactly that of the corresponding Time 1 profiles (labelled by the parameters p1 and p2):

```
MODEL CONSTRAINT:
NEW (pca1 pca2 pca3); ! marginal probabilities for C1
pca1 = EXP(p1) / (EXP(p1) + EXP(p2) + 1);
pca2 = EXP(p2) / (EXP(p1) + EXP(p2) + 1);
pca3 = 1 - (pca1 + pca2);
NEW (p11 p12 p13 p21 p22 p23); !transitions probabilities
p11 = \exp(a1+b11) / (EXP(a1+b11) + EXP(a2+b21) + 1);
p12 = \exp(a1+b12) / (EXP(a1+b12) + EXP(a2+b22) + 1);
p13 = \exp(a1) / (EXP(a1) + EXP(a2) + 1);
p21 = \exp(a2+b21) / (EXP(a1+b11) + EXP(a2+b21) + 1);
p22 = \exp(a2+b22) / (EXP(a1+b12) + EXP(a2+b22) + 1);
p23 = \exp(a2) / (EXP(a1) + EXP(a2) + 1);
NEW (pcb1 pcb2 pcb3);! marginal probabilities for C2
pcb1 = (p11 * pca1) + (p12 * pca2) + (p13 * pca3);
pcb2 = (p21 * pca1) + (p22 * pca2) + (p23 * pca3);
pcb3 = 1 - (pcb1 + pcb2);
! Distributional similarity constraints.
pca1 = pcb1; pca2 = pcb2;
! OR, alternatively:
!\ 0 = pca1 - pcb1;\ 0 = pca2 - pcb2;
```

The computational complexity of this syntax is mainly related to the last section (in red) which serves to constrain the marginal probability of profile membership to equality across time points. Before adding these lines thought, we strongly advocate that the model be first analyzed without them. In other words, starting from the most invariant model from the previous sequence (models a to c), the analyze should be replicated using this expanded syntax, but without imposing distributional similarity constraints. This model should result in the same parameter estimates and model fit information as the previous model, and the class probability provided as part of Mplus output should correspond to those estimated using the new parameters pca1 to pca3 (Time 1) and pcb1 to pcb3 (Time 2). Otherwise, the syntax is likely to involve a typo requiring adjustment.

As we noted before, the computational complexity of this syntax will greatly increase computational time, and make it harder for the model estimation process to converge on a proper solution with a well-replicated maximum likelihood. This type of problem will be reinforced when the model includes transition probabilities that are equal to zero – which requires Mplus to fix some of the multinomial logit parameters representing these transitions. In these situations, the values used by Mplus (in the model estimated without the parts in red) to fix these transitions probabilities to be 0 (referring to the c2#x ON c1#x parameters provided at the end of the results section) can be used to fix these transitions probabilities to 0. This is the relevant part of the output section:

Categorical Latent Variables

C2#1 ON				
C1#1	41.009	0.000	999.000	999.000
C1#2	5.144	0.000	999.000	999.000
C2#2 ON				
C1#1	17.654	0.000	999.000	999.000
C1#2	56.224	0.000	999.000	999.000

However, the marginal probabilities that one seeks to constrain to equality are also a function of these transition probabilities so that forcing some of them to be 0 before formally testing the distributional similarity of the model might influence the end result. For this reason, we do not typically recommend this approach in application where profile membership is highly stable.

For easier referencing, we provide here similar information for LTA including 4-5-6 profiles.

4 profiles

```
%OVERALL%
c2#1 ON c1#1 (b11);
c2#1 ON c1#2 (b12);
c2#1 ON c1#3 (b13);
c2#2 ON c1#1 (b21);
c2#2 ON c1#2 (b22);
c2#2 ON c1#3 (b23);
c2#3 ON c1#1 (b31);
c2#3 ON c1#2 (b32);
c2#3 ON c1#2 (b32);
c2#3 ON c1#2 (b32);
c2#3 ON c1#3 (b33);
[c1#1] (p1); [c1#2] (p2); [c1#3] (p3);
[c2#1] (a1); [c2#2] (a2); [c2#3] (a3);
```

		C2					
		1	2	3	4		
	1	a1 + b11	a2 + b21	a3 + b31	0		
C1	2	a1 + b12	a2 + b22	a3 + b32	0		
	3	a1 + b13	a2 + b23	a3 + b33	0		
	4	a1	a2	a3	0		

```
MODEL CONSTRAINT:
NEW (pca1 pca2 pca3 pca4); ! marginal probabilities for C1
pca1 = EXP(p1) / (EXP(p1) + EXP(p2) + EXP(p3) + 1);
pca2 = EXP(p2) / (EXP(p1) + EXP(p2) + EXP(p3) + 1);
pca3 = EXP(p3) / (EXP(p1) + EXP(p2) + EXP(p3) + 1);
pca4 = 1 - (pca1 + pca2 + pca3);
NEW (p11 p12 p13 p14 p21 p22 p23 p24 p31 p32 p33 p34); !transitions probabilities
p11 = \exp(a1+b11) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + 1);
p12 = \exp(a1+b12) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + 1);
p13 = \exp(a1+b13) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + 1);
p14 = \exp(a1) / (EXP(a1) + EXP(a2) + EXP(a3) + 1);
p21 = \exp(a2+b21) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + 1);
p22 = \exp(a2+b22) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + 1);
p23 = \exp(a2+b23) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + 1);
p24 = \exp(a2) / (EXP(a1) + EXP(a2) + EXP(a3) + 1);
p31 = \exp(a3+b31) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + 1);
p32 = \exp(a3+b32) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + 1);
p33 = \exp(a3+b33) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + 1);
p34 = \exp(a3) / (EXP(a1) + EXP(a2) + EXP(a3) + 1);
NEW (pcb1 pcb2 pcb3 pcb4);! marginal probabilities for C2
pcb1 = (p11 * pca1) + (p12 * pca2) + (p13 * pca3) + (p14 * pca4);
pcb2 = (p21 * pca1) + (p22 * pca2) + (p23 * pca3) + (p24 * pca4);
pcb3 = (p31 * pca1) + (p32 * pca2) + (p33 * pca3) + (p34 * pca4);
pcb4 = 1 - (pcb1 + pcb2 + pcb3);
! Distributional similarity constraints
pca1 = pcb1; pca2 = pcb2; pca3 = pcb3;
!!! Alternatively !!! 0 = pca1 - pcb1; 0 = pca2 - pcb2; 0 = pca3 - pcb3;
```

5 profiles

```
%OVERALL%
c2#1 ON c1#1 (b11);
c2#1 ON c1#2 (b12);
c2#1 ON c1#3 (b13);
c2#1 ON c1#4 (b14);
c2#2 ON c1#1 (b21);
c2#2 ON c1#2 (b22);
c2#2 ON c1#3 (b23);
c2#2 ON c1#4 (b24);
c2#3 ON c1#1 (b31);
c2#3 ON c1#2 (b32);
c2#3 ON c1#3 (b33);
c2#3 ON c1#4 (b34);
c2#4 ON c1#1 (b41);
c2#4 ON c1#2 (b42);
c2#4 ON c1#3 (b43):
c2#4 ON c1#4 (b44);
[c1#1] (p1); [c1#2] (p2); [c1#3] (p3); [c1#4] (p4);
[c2#1] (a1); [c2#2] (a2); [c2#3] (a3); [c2#4] (a4);
```

		C2						
		1	2	3	4	5		
C1	1	a1 + b11	a2 + b21	a3 + b31	a4 + b41	0		
	2	a1 + b12	a2 + b22	a3 + b32	a4+ b42	0		
	3	a1 + b13	a2 + b23	a3 + b33	a4 + b43	0		
	4	a1 + b14	a2 + b24	a3 + b34	a4 + b44	0		
	5	a1	a2	a3	a4	0		

```
MODEL CONSTRAINT:
NEW (pca1 pca2 pca3 pca4 pca5); ! marginal probabilities for C1
pca1 = EXP(p1) / (EXP(p1) + EXP(p2) + EXP(p3) + EXP(p4) + 1);
pca2 = EXP(p2) / (EXP(p1) + EXP(p2) + EXP(p3) + EXP(p4) + 1);
pca3 = EXP(p3) / (EXP(p1) + EXP(p2) + EXP(p3) + EXP(p4) + 1);
pca4 = EXP(p4) / (EXP(p1) + EXP(p2) + EXP(p3) + EXP(p4) + 1);
pca5 = 1 - (pca1 + pca2 + pca3 + pca4);
NEW (p11 p12 p13 p14 p15 p21 p22 p23 p24 p25);
NEW (p31 p32 p33 p34 p35 p41 p42 p43 p44 p45); !transitions probabilities
p11 = \exp(a1+b11) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + EXP(a4+b41) + 1);
p12 = \exp(a1+b12) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + EXP(a4+b42) + 1);
p13 = \exp(a1+b13) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + EXP(a4+b43) + 1);
p14 = \exp(a1+b14) / (EXP(a1+b14) + EXP(a2+b24) + EXP(a3+b34) + EXP(a4+b44) + 1);
p15 = \exp(a1) / (EXP(a1) + EXP(a2) + EXP(a3) + EXP(a4) + 1);
p21 = \exp(a2+b21) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + EXP(a4+b41) + 1);
p22 = \exp(a2+b22) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + EXP(a4+b42) + 1);
p23 = \exp(a2+b23) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + EXP(a4+b43) + 1);
p24 = \exp(a2+b24) / (EXP(a1+b14) + EXP(a2+b24) + EXP(a3+b34) + EXP(a4+b44) + 1);
p25 = \exp(a2) / (EXP(a1) + EXP(a2) + EXP(a3) + EXP(a4) + 1);
p31 = \exp(a3+b31) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + EXP(a4+b41) + 1);
p32 = \exp(a3+b32) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + EXP(a4+b42) + 1):
p33 = \exp(a3+b33) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + EXP(a4+b43) + 1);
p34 = \exp(a3+b34) / (EXP(a1+b14) + EXP(a2+b24) + EXP(a3+b34) + EXP(a4+b44) + 1);
p35 = \exp(a3) / (EXP(a1) + EXP(a2) + EXP(a3) + EXP(a4) + 1);
```

```
p41 = exp (a4+b41) / (EXP(a1+b11) + EXP(a2+b21) + EXP(a3+b31) + EXP(a4+b41) + 1);
p42 = exp (a4+b42) / (EXP(a1+b12) + EXP(a2+b22) + EXP(a3+b32) + EXP(a4+b42) + 1);
p43 = exp (a4+b43) / (EXP(a1+b13) + EXP(a2+b23) + EXP(a3+b33) + EXP(a4+b43) + 1);
p44 = exp (a4+b44) / (EXP(a1+b14) + EXP(a2+b24) + EXP(a3+b34) + EXP(a4+b44) + 1);
p45 = exp (a4) / (EXP(a1) + EXP(a2) + EXP(a3) + EXP(a4) + 1);

NEW (pcb1 pcb2 pcb3 pcb4 pcb5);! marginal probabilities for C2
pcb1 = (p11 * pca1) + (p12 * pca2) + (p13 * pca3) + (p14 * pca4) + (p15 * pca5);
pcb2 = (p21 * pca1) + (p22 * pca2) + (p23 * pca3) + (p24 * pca4) + (p25 * pca5);
pcb3 = (p31 * pca1) + (p32 * pca2) + (p33 * pca3) + (p34 * pca4) + (p35 * pca5);
pcb4 = (p41 * pca1) + (p42 * pca2) + (p43 * pca3) + (p44 * pca4) + (p45 * pca5);
pcb5 = 1 - (pcb1 + pcb2 + pcb3 + pcb4);

! Distributional similarity constraints
pca1 = pcb1; pca2 = pcb2; pca3 = pcb3; pca4 = pcb4;
!!! Alternatively !!! 0=pca1-pcb1;0=pca2-pcb2; 0=pca3-pcb3; 0=pca4-pcb4;
```

6 profiles

```
%OVERALL%
c2#1 ON c1#1 (b11);
c2#1 ON c1#2 (b12);
c2#1 ON c1#3 (b13);
c2#1 ON c1#4 (b14);
c2#1 ON c1#5 (b15);
c2#2 ON c1#1 (b21);
c2#2 ON c1#2 (b22);
c2#2 ON c1#3 (b23);
c2#2 ON c1#4 (b24);
c2#2 ON c1#5 (b25);
c2#3 ON c1#1 (b31);
c2#3 ON c1#2 (b32);
c2#3 ON c1#3 (b33);
c2#3 ON c1#4 (b34);
c2#3 ON c1#5 (b35);
c2#4 ON c1#1 (b41);
c2#4 ON c1#2 (b42);
c2#4 ON c1#3 (b43);
c2#4 ON c1#4 (b44);
c2#4 ON c1#5 (b45);
c2#5 ON c1#1 (b51);
c2#5 ON c1#2 (b52);
c2#5 ON c1#3 (b53);
c2#5 ON c1#4 (b54);
c2#5 ON c1#5 (b55);
[c1#1] (p1);[c1#2] (p2);[c1#3] (p3);[c1#4] (p4);[c1#5](p5);
[c2#1] (a1);[c2#2] (a2);[c2#3] (a3);[c2#4] (a4);[c2#5](a5);
```

	C2						
		1	2	3	4	5	6
	1	a1 + b11	a2 + b21	a3 + b31	a4 + b41	a5 + b51	0
C1	2	a1 + b12	a2 + b22	a3 + b32	a4+ b42	a5+ b52	0
	3	a1 + b13	a2 + b23	a3 + b33	a4 + b43	a5 + b53	0
	4	a1 + b14	a2 + b24	a3 + b34	a4 + b44	a5 + b54	0
	5	a1 + b15	a2 + b25	a3 + b35	a4 + b45	a5 + b55	0
	6	a1	a2	a3	a4	a5	0

```
MODEL CONSTRAINT:
NEW (pca1 pca2 pca3 pca4 pca5 pca6); ! marginal probabilities for C1
pca1=EXP(p1)/(EXP(p1)+EXP(p2)+EXP(p3)+EXP(p4)+EXP(p5)+1);
pca2=EXP(p2)/(EXP(p1)+EXP(p2)+EXP(p3)+EXP(p4)+EXP(p5)+1);
pca3=EXP(p3)/(EXP(p1)+EXP(p2)+EXP(p3)+EXP(p4)+EXP(p5)+1);
pca4=EXP(p4)/(EXP(p1)+EXP(p2)+EXP(p3)+EXP(p4)+EXP(p5)+1);
pca5=EXP(p5)/(EXP(p1)+EXP(p2)+EXP(p3)+EXP(p4)+EXP(p5)+1);
pca6=1-(pca1+pca2+pca3+pca4+pca5);
NEW (p11 p12 p13 p14 p15 p16 p21 p22 p23 p24 p25 p26);
NEW (p31 p32 p33 p34 p35 p36 p41 p42 p43 p44 p45 p46);
NEW (p51 p52 p53 p54 p55 p56);!transitions probabilities
p11=exp(a1+b11)/(EXP(a1+b11)+EXP(a2+b21)+EXP(a3+b31)+EXP(a4+b41)+EXP(a5+b51)+1);
p12=exp(a1+b12)/(EXP(a1+b12)+EXP(a2+b22)+EXP(a3+b32)+EXP(a4+b42)+EXP(a5+b52)+1);
p13=exp(a1+b13)/(EXP(a1+b13)+EXP(a2+b23)+EXP(a3+b33)+EXP(a4+b44)+EXP(a5+b53)+1);
p14=exp(a1+b14)/(EXP(a1+b14)+EXP(a2+b24)+EXP(a3+b34)+EXP(a4+b44)+EXP(a5+b54)+1);
```

```
p15=exp(a1+b15)/(EXP(a1+b15)+EXP(a2+b25)+EXP(a3+b35)+EXP(a4+b45)+EXP(a5+b55)+1);
p16=exp(a1)/(EXP(a1)+EXP(a2)+EXP(a3)+EXP(a4)+EXP(a5)+1);
p21 = \exp(a2+b21)/(EXP(a1+b11)+EXP(a2+b21)+EXP(a3+b31)+EXP(a4+b41)+EXP(a5+b51)+1);
p22=exp(a2+b22)/(EXP(a1+b12)+EXP(a2+b22)+EXP(a3+b32)+EXP(a4+b42)+EXP(a5+b52)+1);
p23 = exp(a2+b23)/(EXP(a1+b13)+EXP(a2+b23)+EXP(a3+b33)+EXP(a4+b43)+EXP(a5+b53)+1);
p24 = \exp(a2 + b24)/(EXP(a1 + b14) + EXP(a2 + b24) + EXP(a3 + b34) + EXP(a4 + b44) + EXP(a5 + b54) + 1);
p25 = exp(a2+b25)/(EXP(a1+b15)+EXP(a2+b25)+EXP(a3+b35)+EXP(a4+b45)+EXP(a5+b55)+1);
p26=exp(a2)/(EXP(a1)+EXP(a2)+EXP(a3)+EXP(a4)+EXP(a5)+1);
p31=exp(a3+b31)/(EXP(a1+b11)+EXP(a2+b21)+EXP(a3+b31)+EXP(a4+b41)+EXP(a5+b51)+1);
p32 = exp(a3+b32)/(EXP(a1+b12)+EXP(a2+b22)+EXP(a3+b32)+EXP(a4+b42)+EXP(a5+b52)+1);
p33 = exp(a3+b33)/(EXP(a1+b13)+EXP(a2+b23)+EXP(a3+b33)+EXP(a4+b43)+EXP(a5+b53)+1);
p34 = \exp(a3 + b34)/(EXP(a1 + b14) + EXP(a2 + b24) + EXP(a3 + b34) + EXP(a4 + b44) + EXP(a5 + b54) + 1);
p35 = exp(a3+b35)/(EXP(a1+b15)+EXP(a2+b25)+EXP(a3+b35)+EXP(a4+b45)+EXP(a5+b55)+1);
p36=exp(a3)/(EXP(a1)+EXP(a2)+EXP(a3)+EXP(a4)+EXP(a5)+1);
p41 = \exp(a4+b41)/(EXP(a1+b11)+EXP(a2+b21)+EXP(a3+b31)+EXP(a4+b41)+EXP(a5+b51)+1);
p42 = \exp(a4+b42)/(EXP(a1+b12)+EXP(a2+b22)+EXP(a3+b32)+EXP(a4+b42)+EXP(a5+b52)+1);
p43 = \exp(a4+b43)/(EXP(a1+b13)+EXP(a2+b23)+EXP(a3+b33)+EXP(a4+b43)+EXP(a5+b53)+1);
p44 = \exp(a4 + b44)/(EXP(a1 + b14) + EXP(a2 + b24) + EXP(a3 + b34) + EXP(a4 + b44) + EXP(a5 + b54) + 1);
p45 = exp(a4+b45)/(EXP(a1+b15)+EXP(a2+b25)+EXP(a3+b35)+EXP(a4+b45)+EXP(a5+b55)+1);
p46=exp(a4)/(EXP(a1)+EXP(a2)+EXP(a3)+EXP(a4)+EXP(a5)+1);
p51=exp(a5+b51)/(EXP(a1+b11)+EXP(a2+b21)+EXP(a3+b31)+EXP(a4+b41)+EXP(a5+b51)+1);
p52 = exp(a5+b52)/(EXP(a1+b12)+EXP(a2+b22)+EXP(a3+b32)+EXP(a4+b42)+EXP(a5+b52)+1);
p53 = exp(a5+b53)/(EXP(a1+b13)+EXP(a2+b23)+EXP(a3+b33)+EXP(a4+b43)+EXP(a5+b53)+1);
p54 = \exp(a5 + b54)/(EXP(a1 + b14) + EXP(a2 + b24) + EXP(a3 + b34) + EXP(a4 + b44) + EXP(a5 + b54) + 1);
p55 = exp(a5 + b55)/(EXP(a1 + b15) + EXP(a2 + b25) + EXP(a3 + b35) + EXP(a4 + b45) + EXP(a5 + b55) + 1);
p56=exp(a5)/(EXP(a1)+EXP(a2)+EXP(a3)+EXP(a4)+EXP(a5)+1);
NEW (pcb1 pcb2 pcb3 pcb4 pcb5 pcb6);! marginal probabilities for C2
pcb1=(p11*pca1)+(p12*pca2)+(p13*pca3)+(p14*pca4)+(p15*pca5)+(p16*pca6);
pcb2=(p21*pca1)+(p22*pca2)+(p23*pca3)+(p24*pca4)+(p25*pca5)+(p26*pca6);
pcb3=(p31*pca1)+(p32*pca2)+(p33*pca3)+(p34*pca4)+(p35*pca5)+(p36*pca6);
pcb4=(p41*pca1)+(p42*pca2)+(p43*pca3)+(p44*pca4)+(p45*pca5)+(p46*pca6);
pcb5=(p51*pca1)+(p52*pca2)+(p53*pca3)+(p54*pca4)+(p55*pca5)+(p56*pca6);
pcb6=1-(pcb1+pcb2+pcb3+pcb4+pcb5);
! Distributional similarity constraints
pca1 = pcb1; pca2 = pcb2; pca3 = pcb3; pca4 = pcb4; pca5 = pcb5;
!!! Alternatively !!! 0=pca1-pcb1;0=pca2-pcb2; 0=pca3-pcb3; 0=pca4-pcb4; 0=pca5-pcb5;
```